

EasyFORM/ENS

**Übernahme von Enscribe-Datenbanken
auf PCs**

Datum: 27.04.2000

**Marcus von Cube
B & S GmbH
Borsigstraße 12
65205 Wiesbaden
Tel: 0 61 22 / 77 05-31
Fax: 0 61 22 / 77 05-25
Email: marcus@mvcsys.de**

Inhaltsverzeichnis

1 Einleitung	1
2 Installation	1
2.1 RemoteEnscribe auf dem Tandem-System.....	1
2.2 Installation auf dem PC.....	2
2.2.1 PC-Programm EF-ENS.exe.....	2
2.2.2 MS-Excel-Arbeitsmappe EF-ENS.xls.....	2
3 Einsatz	4
3.1 Parameter von EF-ENS.exe.....	4
3.1.1 Parameter auf der Kommandozeile.....	4
3.1.2 Parameter als Steuerdatei.....	5
3.2 Bedienung der Excel-Arbeitsmappe.....	6
3.3 Verwendung von EasyFORM mit anderen Anwendungen.....	8
3.3.1 VBA-Makro aus EF-ENS.xls.....	8
3.3.2 Beispiel einer Ausgabedatei.....	11

1 Einleitung

EasyFORM/ENS ist eine Anwendung der RemoteEnscribe-Technologie zur Übernahme von Tandem Enscribe-Datenbanken auf PCs. EasyFORM ist ein externes Programm, daß auf dem PC läuft und über das Netzwerk Datenbank-Befehle an das Tandem-System sendet. Um dem Anwender den Umgang zu erleichtern gehört eine MS-Excel-Arbeitsmappe zum Lieferumfang, die die Funktionen durch einen Dialog zugänglich macht und die Ergebnisse direkt in eine Seite des Arbeitsblattes kopiert.

2 Installation

2.1 RemoteEnscribe auf dem Tandem-System

Um EasyFORM benutzen zu können, ist zuerst *RemoteEnscribe* auf dem Tandem-System zu installieren:

1. Kopieren Sie RSQLSRV.BIN¹ als Binärdatei auf die Tandem und geben Sie ihr den Filecode 100. Das kann mit folgenden FTP-Befehlen auf dem PC erreicht werden:

```
ftp tandem
(Benutzer und Kennwort)
bin
put RSQLSRV.BIN $SYSTEM.UTIL.RSQLSRV,100
```

2. RSQLSRV kann nun auf der Tandem mit SQLCOMP compiliert werden:

```
SQLCOMP / IN $SYSTEM.UTIL.RSQLSRV / CATALOG $VOLUME.SQLCAT
```

Geben Sie einen beliebigen SQL-Katalog auf ihrem System an. Für EasyFORM/ENS ist dieser Schritt nicht unbedingt notwendig.

3. Damit RSQLSRV automatisch gestartet wird, wenn der PC eine Anfrage macht, muß noch die TCP/IP-Konfiguration erweitert werden. Editieren Sie dazu die Datei PORTCONF in ihrem TCP/IP-Konfigurations-Subvolume auf dem Tandem-System. Fügen sei eine Zeile nach folgendem Muster hinzu:

```
741 $SYSTEM.UTIL.RSQLSRV
```

Die Zahl 741 ist die TCP-Portnummer und kann von Ihnen auch anders gewählt werden.

4. Damit die Konfiguration aktiv wird, muß der LISTNER-Prozess neu gestartet werden:

```
STOP $LSN0
LISTNER / NAME $LSN0, PRI 160, TERM $TM01.#A / $SYSTEM.TCPIP.PORTCONF
```

Geben Sie die für Sie zutreffenden Parameter und den Namen Ihrer PORTCONF-Datei an.

Nach erfolgreicher Installation können Sie nun PC-Anwendungen, die *RemoteEnscribe* benutzen, starten. Geben Sie beim Aufruf dieser Programme bitte die Portnummer, die Sie hier in PORTCONF eingetragen haben, an. Der Default der meisten dieser Anwendungen ist 741.

¹ RSQLSERV ist eigentlich der *RemoteSQL*-Server, der aber die Funktionalität von *RemoteEnscribe* enthält. Für die vorliegende Anwendung ist es nicht erforderlich, SQL zu installieren oder den Server mit SQLCOMP zu bearbeiten.


2.2 Installation auf dem PC

2.2.1 PC-Programm EF-ENS.exe

Kopieren sie EF-ENS.exe in ein beliebiges Verzeichnis auf dem PC oder im Netzwerk. Die aktuelle Version ist als 32-Bit-Programm realisiert. Eine 16-Bit Version, die auch unter Windows 3.x oder mit der 16-Bit-Version von Excel funktioniert, ist auf Anfrage verfügbar¹.

2.2.2 MS-Excel-Arbeitsmappe EF-ENS.xls

Die mitgelieferte MS-Excel-Arbeitsmappe muß noch eingerichtet werden, damit die Vorgaben für Ihre Installation passen. Starten Sie dazu MS-Excel und öffnen Sie die Arbeitmappe. Wechseln Sie auf das Blatt *Parameter*:



Für die Installation sind die Felder *Programm* und *Datei* interessant:

1. Im Feld *Programm* tragen sie die Position von EF-ENS.exe auf Ihrem Rechner ein.
2. Im Feld *Datei* tragen sie den Namen einer Zwischendatei für die Ausgabe der Tandemdaten ein. Die Datei wird von EF-ENS.exe geschrieben und dann vom Excel-Makro in die Arbeitsmappe kopiert.
3. Im Feld *Komma* können sie das Dezimalkommazeichen "." oder "," eintragen. Änderungen sind nur notwendig, wenn der Import von Zahlen nicht korrekt funktioniert, was mit den Einstellungen Ihrer Excel-Version zusammenhängen kann.
4. Im Feld *Anzahl* kann für Testzwecke die Maximalzahl der gelesenen Datensätze eingeschränkt werden.

¹ Versionen für andere Betriebssysteme wie OS/2 oder Linux, dann allerdings ohne die Unterstützung durch eine Arbeitsmappe, sind auch verfügbar.

5. Das Feld *Namenstrenner* setzt das Zeichen, das zusammengesetzte Feldnamen der Form *RECORD.FIELD* trennt.
6. Die übrigen Felder werden durch den Dialog gesetzt, können aber auch hier vorbelegt werden.
Die geänderte Arbeitsmappe installieren Sie am besten als Vorlage EasyFORM.xlt in Ihrem Excel-Verzeichnis.

3 Einsatz

3.1 Parameter von EF-ENS.exe

EF-ENS.exe muß mit Kommandozeilen-Parametern versorgt werden, damit es arbeitet. Starten Sie das Programm ohne Parameter gibt es eine kurze Anleitung aus:

```
EasyFORM/ENS - Zugriff auf ENSCRIBE-Datenbanken - (c) 1997-2000 Marcus von Cube

EF-ENS [Optionen] <Zieldatei> <Host>[:<Port>] <User> "<Pass>"
        <Dict> <Record> [<Quelldatei>]
Optionen:
-n<Namenstiefe>: Anzahl der Ebenen pro Spaltenname
-d<Aufloesung>:  Auflösen nur bis zu diesem Level
-c<Anzahl>:      Maximale Anzahl von Saetzen
-(<Key>,"<Start>","<Ende>"): Schluesselwahl
Alternative: EF-ENS @<Steuerdatei>
              In der Steuerdatei stehen die Argumente zeilenweise ohne ""
Als weitere Parameter koennen folgen:
<Dezimalpunkt> (. oder ,)
<Text-Anfang>  (")
<Text-Ende>    (")
<Namenstrenner> (.)
```

3.1.1 Parameter auf der Kommandozeile

Die Parameter können direkt auf der Kommandozeile eingegeben werden. Die Parameter im einzelnen:

1. Folgende *Optionen* steuern die Arbeit des Programms. Sie können in beliebiger Reihenfolge angegeben werden, stehen aber vor allen anderen Parametern:
 1. *-nZahl*: Die eingegebene Zahl steuert die Anzahl der Ebenen von verschachtelten Namen, die für die Überschriften erzeugt werden. Ohne diesen Parameter werden alle Ebenen aufgelöst. Ein Wert von 1 gibt nur den Basisnamen jeder Spalte aus. Ein negativer Wert steuert die Namensauflösung von links aus, benennt also die oberste zu erzeugende Ebene. Default ist -1.
 2. *-dZahl*: Die eingegebene Zahl steuert die Anzahl der Ebenen von verschachtelten Feldern, bis zu denen die Spalten aufgelöst werden. Man kann so verhindern, das Spalten wie DATUM, die intern noch als JAHR/MONAT/TAG definiert sind, auseinandergerissen werden.
 3. *-cZahl*: Es werden nur maximal *Zahl* Datensätze gelesen.
 4. *-("Key", "Von", "Bis")*: Die Datensätze werden nach dem Schlüssel *Key* gelesen. Wenn der Schlüssel (Key-Tag) als Zeichenkette definiert wurde, muß er in Anführungszeichen stehen. 0 ist der Primärschlüssel einer KEY-SEQUENCED-Datei. Als Vergleichswerte sind nur Zeichenketten zugelassen. Sie können auch weggelassen werden.
2. *Zieldatei* bezeichnet den Namen einer Datei, in der EasyFORM die Ergebnisse der Abfrage speichert. Die Daten werden als Delimited-ASCII erzeugt. Das Trennzeichen ist ein Tabulator. Texte werden in Anführungszeichen eingeschlossen, wenn in den entsprechenden Parametern nichts anderes angegeben wurde. Die erste Zeile der Datei enthält die Spaltennamen.

TIP: Endet der Name der Zieldatei auf "HTM" oder "HTML" (groß oder klein), dann wird eine Datei im HTML-Format erzeugt, die sich direkt mit einem Internet-Browser anzeigen und ausdrucken läßt.

3. *Host[:Port]* ist die TCP/IP-Adresse des Tandem-Systems. Die Portnummer hinter dem Doppelpunkt kann angegeben werden, wenn sie sich von 741 unterscheidet (siehe Abschnitt 2.1 *RemoteEnscribe auf dem Tandem-System.*)
4. *User* ist ein gültiger Tandem-Benutzername im Klartext.
5. *Pass* ist das zugehörige Passwort. Wenn das Passwort leer ist, dann geben Sie bitte "" ein.
6. *Dict* ist ein Subvolume mit einem DDL-Dictionary.
7. *Record* ist ein *RECORD* aus dem genannten Dictionary.
8. Die *Quelldatei* auf dem Tandem-System kann angegeben werden, wenn sie sich von der Vorgabe aus dem Dictionary unterscheidet oder die Angaben unvollständig sind (kein SUBVOLUME). Als Default kann ein "*" eingegeben werden, wenn die folgenden Parameter benötigt werden.
9. *Dezimalpunkt* ist entweder ein Punkt oder ein Komma und wird benutzt, um skalierte numerische Werte zu formatieren. Default ist ein Punkt.
10. *Text-Anfang* und *Text-Ende* sind Delimiter für Textwerte. Default ist ein doppeltes Anführungszeichen. Es können beliebige -auch leere- Strings angegeben werden. Wird *Text-Ende* nicht angegeben, richtet es sich nach *Text-Anfang*. Sie müssen die beiden Parameter in Anführungszeichen einschließen, wenn sie Sonderzeichen enthalten.
11. *Namenstrenner* trennt Bestandteile zusammengesetzter Namen, wie sie bei verschachtelten Definitionen vorkommen.

3.1.2 Parameter als Steuerdatei

EasyFORM kann auch mit dem Namen einer Steuerdatei hinter dem Zeichen @ aufgerufen werden. In dieser Datei stehen die Parameter Zeile für Zeile und ohne die begrenzenden Anführungszeichen. Dies ist die Form, die auch das Excel-Makro benutzt. Als Parameterdatei wird vom Makro der gleiche Name wie die Ausgabedatei benutzt. Bricht die Verarbeitung ab, dann enthält die Ausgabedatei also keine Daten sondern die Aufrufparameter!

Beispiel einer Steuerdatei:


```
-c20
-n2
-d1
-( 'KT' , "AAAA" , "Z" )
test.html
tandem:741
dev.marcus
kennwort
$data.mcdict
test
mcdcl.testfile
,
"
"
.
```

TIP: Nennen Sie die Eingabedateien *irgendwas.efe* und verknüpfen sie die Dateierweiterung *efe* mit dem Programm EF-ENS.exe, dann können Sie Anfragen durch Anklicken des Icons der Steuerdatei starten.


3.2 Bedienung der Excel-Arbeitsmappe

Um EasyFORM über die Excel-Arbeitsmappe zu bedienen, gehen Sie wie folgt vor:

1. Öffnen Sie die Arbeitsmappe oder erstellen Sie aus der Vorlage EasyFORM.xlt eine neue. Wechseln Sie auf das Blatt *Start*:




2. Klicken Sie auf *Neu*, um eine Abfrage zu formulieren. Es erscheint der folgende Dialog:



System	<input type="text" value="tandem"/>
User	<input type="text" value="dev.marcus"/>
Password	<input type="text" value="*****"/>
Dictionary	<input type="text" value="\$data.tgdict"/>
Record	<input type="text" value="TD037"/>
Datei	<input type="text" value="*"/>
Key	<input f6\""="" type="text" value="\"/> Benutzen Sie Anführungszeichen für diese drei Felder
Von/Bis	<input 030000\""="" type="text" value="\"/> <input 030500\""="" type="text" value="\"/>
Auflösungstiefe	<input type="text" value="2"/> Namensebenen <input type="text" value="2"/>
<input type="button" value="OK"/> <input type="button" value="Abbrechen"/>	

3. Füllen Sie die Felder aus. Aus den Angaben wird eine Steuerdatei erstellt.

4. Senden Sie die Dialogbox mit OK ab. Es wird das Programm EF-ENS.exe in einen eigenen Fenster gestartet. Der folgende Bildschirm zeigt die Ausgaben eines erfolgreichen Programmlaufs:




```
D:\EasyFORM\EF-ENS.exe
EasyFORM/ENS - Zugriff auf ENSCRIBE-Datenbanken - (c) 1997-2000 Marcus von Cube

Verbindungsaufbau zu tandem:740...OK
Lese Record TD037 aus Dictionary $DATA.TGDICT...OK
Deffne die Eingabedatei TGDATA.TD037...OK
Deffne die Ausgabedatei D:\EasyFORM\EasyFORM.txt...OK
Schlüsselspalte ist FILIALE-6

Anzahl der gelesenen Datensätze: 74

Datei D:\EasyFORM\EasyFORM.txt wurde erzeugt
ENTER..._
```

5. Achten Sie auf Fehlermeldungen, die ebenfalls in diesem Fenster ausgegeben werden. Wenn Die Abfrage fehlerfrei durchgeführt wurde, können Sie das Fenster schließen. Es erscheint wieder Excel mit folgender Abfrage:



6. Wenn EasyFORM korrekt durchgeführt wurde, können Sie den Dialog mit OK bestätigen. Das Excelmakro importiert dann die Zwischendatei in die Seite *Daten* und zeigt diese an. Sie können nun eine neue Abfrage starten oder die gelesenen Daten mit Excel weiter bearbeiten. Speichern Sie ihre Ergebnisse in einer neuen Arbeitsmappe ab. Diese können sie jederzeit wieder benutzen, um die Abfrage zu wiederholen oder zu ändern.
7. Mit der Taste *Anzeigen* des Startbildschirms wechseln Sie auf die Seite *Daten*.
8. Mit der Taste *Löschen* des Startbildschirms löschen Sie den Inhalt der *Daten*-Seite. Dies ist sinnvoll, wenn Sie zwar die Abfrage, nicht aber ihr vollständiges Ergebnis (z. B. wegen der Größe) mit der Arbeitsmappe speichern wollen.

3.3 Verwendung von EasyFORM mit anderen Anwendungen

Andere Anwendungen, zum Beispiel Datenbanken, können EasyFORM ähnlich benutzen wie Excel. Um eine Hilfe für ihrer eigene Makroprogrammierung zu bieten folgt hier der Text des VBA-Makros, das die Excel-Arbeitsmappe steuert.

3.3.1 VBA-Makro aus EF-ENS.xls

```

'
' EasyFORM_ENS
'
' MvC 2000
'
' Aufgabe: Zugriff auf Tandem-Enscribe-Daten und Einarbeitung in eine
'          EXCEL-Tabelle
'
' Benötigt: EasyFORM/ENS und Remote-Enscribe
'
' Datum      Vers.  Bearb.   Erläuterung
'-----
' 26.04.00   1.0    MvC      erstellt
'
Option Explicit

Dim Dialog As DialogSheet
Dim Parameter As Worksheet
Dim Daten As Worksheet
Dim OK As Boolean
Dim Datei As String

Sub EasyFORM_ENS()
    Set Dialog = DialogSheets("Dialog")
    Set Parameter = Worksheets("Parameter")
    Set Daten = Worksheets("Daten")
    Datei = Parameter.Range("Datei").Value

    Dialog.Buttons("OK").DismissButton = True
    Dialog.Show
    If Not OK Then
        MsgBox "Abgebrochen", 0, "EasyFORM"
        Exit Sub
    End If

    Dim cmd As String
    cmd = Parameter.Range("Programm").Value
    Dim i As Integer
    Dim p As String

    '
    ' Parameter ueber Steuer-Datei
    '
    cmd = cmd + " @" + Datei

    Open Datei For Output As #1
    p = Parameter.Range("Anzahl").Value
    If p <> "" Then Print #1, "-c" + p
    p = Parameter.Range("Tiefe").Value
    If p <> "" Then Print #1, "-d" + p
    p = Parameter.Range("Level").Value
    If p <> "" Then Print #1, "-n" + p
    p = Parameter.Range("Key").Value

```

```
If p <> "" Then
    Print #1, "-" + p;
    p = Parameter.Range("Von").Value
    If p <> "" Then
        Print #1, "," + p;
    Else
        Print #1, " ,"""";
    End If
    p = Parameter.Range("Bis").Value
    If p <> "" Then
        Print #1, "," + p;
    End If
    Print #1, ")" + p
End If
Print #1, Datei
Print #1, Parameter.Range("System").Value
Print #1, Parameter.Range("User").Value
Print #1, Parameter.Range("Password").Value
Print #1, Parameter.Range("Dictionary").Value
Print #1, Parameter.Range("Record").Value
p = Parameter.Range("InputFile").Value
If p <> "" Then
    Print #1, p
Else
    Print #1, "*"
End If
Print #1, Parameter.Range("Komma").Value
Print #1, """"
Print #1, """"
Print #1, Parameter.Range("Trenner").Value
Close #1

'MsgBox cmd
Shell cmd, 1

If vbOK = MsgBox("Drücken Sie OK, um die Zwischendatei zu übernehmen", _
                vbOKCancel, "EasyFORM") _
Then
    Import
End If

End Sub

'
' Import der Text-Datei
'
Sub Import()
    Dim ImportFile As Workbook
    Daten.Cells.Delete
    Workbooks.OpenText Filename:=Datei, Tab:=True
    Set ImportFile = ActiveWorkbook
    ActiveSheet.Cells.Copy destination:=Daten.Cells
    ImportFile.Close
    Daten.Activate
    'Cells.Select
    Cells.EntireColumn.AutoFit
    ActiveSheet.Cells(2, 1).Select
End Sub

'
' Dialog-Funktionen
'
Sub Dialog_BeiAnzeigen()
    GetParam ("System")
    GetParam ("User")
    GetParam ("Password")
    GetParam ("Dictionary")
    GetParam ("Record")
    GetParam ("InputFile")
    GetParam ("Key")
```

```
    GetParam ("Von")
    GetParam ("Bis")
    GetParam ("Tiefe")
    GetParam ("Level")
    Dialog.Focus = "D_System"
End Sub

Sub Abbrechen_BeiKlick()
    OK = False
End Sub

Sub OK_BeiKlick()
    If IsEmpty("System") Then Exit Sub
    If IsEmpty("User") Then Exit Sub
    If IsEmpty("Dictionary") Then Exit Sub
    SetParam ("System")
    SetParam ("User")
    SetParam ("Password")
    SetParam ("Dictionary")
    SetParam ("Record")
    SetParam ("InputFile")
    SetParam ("Key")
    SetParam ("Von")
    SetParam ("Bis")
    SetParam ("Tiefe")
    SetParam ("Level")

    OK = True
End Sub

Function IsEmpty(feld As String) As Boolean
    Dim d_feld As String
    d_feld = "D_" + feld
    With Dialog
        If .EditBoxes(d_feld).Text = "" Then
            .Focus = d_feld
            MsgBox feld + " muß eingegeben werden"
            .Buttons("OK").DismissButton = False
            IsEmpty = True
        Else
            IsEmpty = False
        End If
    End With
End Function

Sub GetParam(feld As String)
    Dialog.EditBoxes("D_" + feld).Text = Parameter.Range(feld).Value
End Sub

Sub SetParam(feld As String)
    Parameter.Range(feld).Value = Dialog.EditBoxes("D_" + feld).Text
End Sub

Sub Anzeigen()
    Worksheets("Daten").Activate
End Sub

Sub Loeschen()
    Worksheets("Daten").Cells.Delete
End Sub
```

3.3.2 Beispiel einer Ausgabedatei

PERSON	NAME	BEARBEITER	DATUM
"710640"	"Schuchna, Tanja"	"705574"	20000427
"710715"	"Hoefner, Manuela"	"715101"	20000427
"710962"	"Weinzierl, Antigo"	"705574"	20000427
"711028"	"Synde, Gabriele"	"705574"	20000427
"711085"	"Christopf, Belinda"	"711028"	20000427
"711739"	"Thiemig, Maringa"	"711028"	20000427
"713214"	"Roesch, Margarete"	"703595"	20000427
"715101"	"Edenharder, Sabine"	"742876"	20000427
"715821"	"Pirkl, Christa"	"742876"	20000427
"729087"	"Meier, Birgit"	"742876"	20000427
"736421"	"Kajgana, Sadeta"	"742551"	20000427
"738369"	"Catalkava, Fatima"	"742551"	20000427
"740126"	"Zweck, Gerlinde"	"742551"	20000427
"740738"	"Bierschneider, Anne"	"742551"	20000427
"741454"	"Baumer, Gertrud"	"742551"	20000427
"741504"	"Hofbauer, Elisabeth"	"715101"	20000427
"742403"	"Werner, Annemarie"	"715101"	20000427
"742551"	"Nusselt, Rita"	"715101"	20000427
"742876"	"Hofmann, Andrea"	"715101"	20000427
"763847"	"Strauch, Sabine"	"742876"	20000427

Die Spalten sind mit Tabulatoren getrennt. Die Trennzeichen um Zeichenketten sind beim Aufruf von EasyFORM einstellbar (siehe 3.1.1 Parameter auf der Kommandozeile.) Die in diesem Beispiel verwendeten Zahlen sind in der Dateibeschreibung als Strings vereinbart und daher ebenfalls in Anführungszeichen eingeschlossen. Excel kümmert sich darum allerdings nicht und verwandelt sie trotzdem in Zahlen.

Skalierte Zahlen werden mit dem eingestellten Dezimalkommazeichen, normalerweise ein Punkt, ausgegeben.